Sub Заменить_ссылки_на_книгу_константами_в_активной_книге()
    Dim Sh As Worksheet, c As Range, MyFileName$, Arr
    With Application.FileDialog(msoFileDialogFilePicker)
        .Title = "Укажите файл, связи на который будут заменены константами в формулах:"
        .InitialFileName = ActiveWorkbook.Path & "\"
        .AllowMultiSelect = False
        .ButtonName = "OK"
        .Filters.Add Description:="Файлы Microsoft Excel", _
            Extensions:="*.xls; *.xl*"
        If .Show = 0 Then
            Exit Sub
        End If
        MyFileName = .SelectedItems(1)
    End With
    Arr = Split(MyFileName, "\")
    MyFileName = "[" & Arr(UBound(Arr, 1)) & "]"
    For Each Sh In ActiveWorkbook.Worksheets
        For Each c In Sh.UsedRange.Cells
            If c.HasFormula Then
                Call Links_To_Constants2(c, MyFileName)
            End If
        Next c
    Next Sh
End Sub
 
Sub Links_To_Constants2(CLL As Range, MyFileName$)
    Dim FormulaOld$, FormulaNew$, s$, sNum$, sTmp$, sLink$, LinkSuffix$, LinkSuffix2$
    Dim LinkNumber, i%, j%, k%, n%, r1&, r2&, c1&, c2&, rOffset&, cOffset&, Status%, LenFormula%
    Dim Usl As Boolean, Usl2 As Boolean, Sboy As Boolean, UslArray As Boolean, _
    UslOpen  As Boolean, UslOpen2 As Boolean, UslR As Boolean, UslC As Boolean
    FormulaOld = CLL.FormulaR1C1 ' Стиль формул R1C1
    FormulaNew = ""
    If InStr(1, FormulaOld, MyFileName) > 0 Then
        LenFormula = Len(FormulaOld)
        For i = 1 To LenFormula
            sTmp = Mid(FormulaOld, i, 1)
            Select Case sTmp
                 Case "'"
                    If Status = 0 Then
                        Usl = True
                        Status = 1
                    End If
                 Case "["
                    If Not Usl Then
                        If Status = 0 Then Status = 1
                    End If
                    If Status > 3 Then
                        UslOpen = True
                    End If
                 Case "."
                    If Status = 1 Then
                        Status = 2
                    End If
                 Case "]"
                    If Status = 2 Then
                        Status = 3
                    Else
                        Sboy = True
                    End If
                    UslOpen = False
                 Case "!"
                    If Status = 3 Then
                        Status = 4
                    Else
                        Sboy = True
                    End If
                 Case "R"
                    If Status = 4 Then
                        Status = 5
                    End If
                 Case "C"
                    If Status = 5 Then
                        Status = 6
                    End If
                 Case ":"
                    If Status = 6 Then
                        UslArray = True
                    End If
                 Case "-"
                    If Not UslOpen Then
                        If Status = 6 Then
                            Status = 7
                        ElseIf Status > 4 Then
                            Sboy = True
                        End If
                    End If
                 Case " ", "+", "/", "*", "^", "(", ")", ",", ";"
                    If Status = 6 Then
                        Status = 7
                    ElseIf Status > 4 Then
                        Sboy = True
                    End If
            End Select
            If Not Sboy Then
                If i = LenFormula Then
                    If Status = 6 Then
                        LinkSuffix = LinkSuffix & sTmp
                        sTmp = ""
                        Status = 7
                    End If
                End If
                If Status = 7 Then
                    sNum = ""
                    LinkSuffix2 = ""
                    r1 = 0: r2 = 0: c1 = 0: c2 = 0
                    For j = 1 To Len(LinkSuffix)
                        s = Mid(LinkSuffix, j, 1)
                        Select Case s
                            Case "["
                                UslOpen2 = True
                            Case "]"
                                UslOpen2 = False
                                If UslR Then
                                    LinkSuffix2 = LinkSuffix2 & Format(Val(sNum) + CLL.row)
                                    If r1 = 0 Then
                                       r1 = Val(sNum) + CLL.row
                                    Else
                                       r2 = Val(sNum) + CLL.row
                                    End If
                                    UslR = False
                                ElseIf UslC Then
                                    LinkSuffix2 = LinkSuffix2 & Format(Val(sNum) + CLL.Column)
                                    If c1 = 0 Then
                                       c1 = Val(sNum) + CLL.Column
                                    Else
                                       c2 = Val(sNum) + CLL.Column
                                    End If
                                    UslC = False
                                End If
                                sNum = ""
                            Case "R"
                                UslR = True
                                LinkSuffix2 = LinkSuffix2 & s
                            Case "C"
                                UslC = True
                                LinkSuffix2 = LinkSuffix2 & s
                            Case ":"
                                LinkSuffix2 = LinkSuffix2 & s
                            Case Else
                                sNum = sNum & s
                                If Not UslOpen2 Then
                                    LinkSuffix2 = LinkSuffix2 & s
                                End If
                        End Select
                        If sNum <> "" Then
                            Usl2 = False
                            If j = Len(LinkSuffix) Then
                                Usl2 = True
                            Else
                                Select Case s
                                    Case "R", "C", ":"
                                        Usl2 = True
                                End Select
                            End If
                            If UslArray And Usl2 Then
                                If UslR Then
                                    If r1 = 0 Then
                                       r1 = Val(sNum)
                                    Else
                                       r2 = Val(sNum)
                                    End If
                                    UslR = False
                                ElseIf UslC Then
                                    If c1 = 0 Then
                                       c1 = Val(sNum)
                                    Else
                                       c2 = Val(sNum)
                                    End If
                                    UslC = False
                                End If
                                sNum = ""
                            End If
                        End If
                    Next j
                    LinkSuffix = LinkSuffix2
                    If LinkSuffix Like "RC:R*" Then
                        If r2 = 0 Then r2 = r1
                        If c2 = 0 Then c2 = c1
                        r1 = CLL.row: c1 = CLL.Column
                    ElseIf LinkSuffix Like "R*C:*" Then
                        If c2 = 0 Then c2 = c1
                        c1 = CLL.Column
                    ElseIf LinkSuffix Like "RC*:*" Then
                        If r2 = 0 Then r2 = r1
                        r1 = CLL.row
                    End If
                    If LinkSuffix Like "*:RC" Then
                        r2 = CLL.row: c2 = CLL.Column
                    ElseIf LinkSuffix Like "*:R*C" Then
                        c2 = CLL.Column
                    ElseIf LinkSuffix Like "*:RC*" Then
                        r2 = CLL.row
                    End If
                    If UslArray Then
                        FormulaNew = FormulaNew & "{"
                        If r2 < r1 Then k = r2: r1 = r2: r2 = k
                        If c2 < c1 Then k = c2: c1 = c2: c2 = k
                        For k = r1 To r2
                            For n = c1 To c2
                                s = sLink & "R" & Format(k) & "C" & Format(n)
                                LinkNumber = ExecuteExcel4Macro(s)
                                If IsNumeric(LinkNumber) Then
                                    LinkNumber = Replace(LinkNumber, ",", ".")
                                Else
                                    LinkNumber = """" & LinkNumber & """"
                                End If
                                FormulaNew = FormulaNew & LinkNumber
                                If Not (k = r2 And n = c2) Then
                                    FormulaNew = FormulaNew & ","
                                End If
                            Next n
                        Next k
                        FormulaNew = FormulaNew & "}"
                        sLink = ""
                    Else
                        If LinkSuffix = "RC" Then
                            LinkSuffix = "R" & Format(CLL.row) & "C" & Format(CLL.Column)
                        ElseIf LinkSuffix Like "R*C" Then
                            LinkSuffix = Replace(LinkSuffix, "C", "C" & Format(CLL.Column))
                        ElseIf LinkSuffix Like "RC*" Then
                            LinkSuffix = Replace(LinkSuffix, "R", "R" & Format(CLL.row))
                        End If
                        sLink = sLink & LinkSuffix
                        LinkNumber = ExecuteExcel4Macro(sLink)
                        If IsNumeric(LinkNumber) Then
                            LinkNumber = Replace(LinkNumber, ",", ".")
                        Else
                            LinkNumber = """" & LinkNumber & """"
                        End If
                        FormulaNew = FormulaNew & LinkNumber
                        sLink = ""
                    End If
                    Sboy = True
                ElseIf Status = 0 Then
                    FormulaNew = FormulaNew & sTmp
                ElseIf Status < 4 Then
                    sLink = sLink & sTmp
                ElseIf Status = 4 Then
                    If InStr(1, sLink, MyFileName) > 0 Then
                        sLink = sLink & sTmp
                    Else
                        Sboy = True
                    End If
                Else
                    LinkSuffix = LinkSuffix & sTmp
                End If
            End If
            If Sboy Then
                FormulaNew = FormulaNew & sLink & sTmp
                Usl = False
                UslArray = False
                Sboy = False
                Status = 0
                sLink = ""
                LinkSuffix = ""
            End If
        Next i
    Else
        FormulaNew = FormulaOld
    End If
    If Trim(FormulaNew) = "=" Then FormulaNew = ""
    CLL.FormulaR1C1 = FormulaNew
End Sub

